BusinessObjects
Tips and Tricks

Presented by
Brie Anne Clark

CREATIVE TECHNOLOGY
& TRAINING SOLUTIONS

http://www.cttsbi.com

Revenue Growth Report

Problem:

I want a report that shows the growth in sales revenue per quarter since last year, but I want BUSINESSOBJECTS to determine which years to use based on the current date, filter the report accordingly, and then calculate the variance.

Solution:

1. Create a report using **Year**, **Quarter**, and **Sales revenue** from the **eFashion** universe.

2. Set **Quarter** as the master cell, and then manually insert a variance that subtracts last year from this year. Also insert a percentage variance between the two years. Now you have values to which you can compare the variables that you create.

Q1

Year	Sales revenue
2002	$2,660,700
2003	$3,326,172
2003-2002	$665,473
2003-2002	25.01 %
2004	$3,742,989

3. Next, create a variable named **Last Year Flag** to flag last year.

```
=If <Year>= FormatNumber((Year(CurrentDate())-1),"0000") Then 1
```

   Also create a variable named **Current Year Flag** to flag this year.

```
=If <Year>= FormatNumber((Year(CurrentDate())),"0000") Then 1
```

4. Now that there are flags for the previous and current years, create a variable named **Quarterly Variance** that subtracts last year's revenue from this year's revenue.

```
=<Sales revenue> Where (<Last Year Flag>=1) - <Sales revenue> Where
(<Current Year Flag>=1)
```

   Also create a variable named **Quarterly Variance %** that subtracts last year's revenue from this year's revenue, and then divides that value by last year's revenue.

```
=(<Sales revenue> Where (<Last Year Flag>=1) - <Sales revenue> Where
(<Current Year Flag>=1))/<Sales revenue> Where (<Last Year Flag=1)
```

5. Duplicate the sectioned report with the variance. Delete the variance rows from the block. Then, add a sum and an average for **Sales revenue** to create rows to hold the **Quarterly Variance** and **Quarterly Variance %** variables.

Q1

Year	Sales revenue
2002	$2,660,700
2003	$3,326,172
2004	$3,742,989
Sum:	**$9,729,861**
Average:	**$3,243,287**

6. Drag **Quarterly Variance** and **Quarterly Variance %** from the report manager to replace the sum and average you inserted. Format the **Quarterly Variance** cell as currency and format the **Quarterly Variance %** cell to display as a percentage.

Q1

Year	Sales revenue
2002	$2,660,700
2003	$3,326,172
2004	$3,742,989
Sum:	**$665,472.70**
Average:	**25.01 %**

7. Next, create a variable named **Variance Label** to label the new variables.

```
=FormatNumber((Year(CurrentDate())) ,"0000") + "-" +
FormatNumber((Year(CurrentDate())-1) ,"0000")
```

8. Drag **Variance Label** from the report manager to replace the **Sum** and **Average** labels.

Q1

Year	Sales revenue
2002	$2,660,700
2003	$3,326,172
2004	$3,742,989
2003-2002	**665,472.70**
2003-2002	**25.01 %**

9. Finally, add a filter to the block to show only last year and this year. To do this, click on the block, and then select **Filters** from the **Format** menu. In the **Format Filters** dialog box, select the **Table 1** folder, click **Add**, select **Year**, and then click **OK**.

10. Next, select the **Year** filter under the **Table 1** folder and click **Define**. Enter the following formula and click **OK**.

```
=<Current Year Flag>=1 Or <Last Year Flag>=1
```

11. Click **OK** to close the Format Filters dialog box and you should see the following report.

Q1

Year	Sales revenue
2002	$2,660,700
2003	$3,326,172
2003-2002	**665,472.70**
2003-2002	**25.01 %**

Q2

Year	Sales revenue
2002	$2,278,693
2003	$2,840,651
2003-2002	**561,957.40**
2003-2002	**24.66 %**

Q3

Year	Sales revenue
2002	$1,367,841
2003	$2,879,303
2003-2002	**1,511,462.30**
2003-2002	**110.50 %**

Q4

Year	Sales revenue
2002	$1,788,580
2003	$4,186,120
2003-2002	**2,397,539.60**
2003-2002	**134.05 %**

Past Six Months Summary Report

Problem:

I only want to display data from the first day of the month six months ago through the current date.

Solution:

1. Create a new report using **Sales Person**, **Invoice Date**, and **Revenue** from the **Island Resorts Marketing** universe.

2. Create a measure variable named **Six Months Ago** that subtracts six from the current month, and if that result is less than or equal to zero, it adds twelve. If the current month minus six is greater than zero, it returns that value.

   ```
   =If MonthNumberOfYear(CurrentDate())-6<=0 Then
   MonthNumberOfYear(CurrentDate())+6 Else
   MonthNumberOfYear(CurrentDate())-6
   ```

3. Next, create a measure variable named **Year of Six Months Ago** that checks to see if the current month minus six is less than or equal to zero. If it is, then it returns the current year minus one. If it is a positive number, it simply returns the current year.

   ```
   =If MonthNumberOfYear(CurrentDate())-6 <=0 Then Year(CurrentDate())-
   1 Else Year(CurrentDate())
   ```

4. Create a dimension variable named **First Day of Six Months Ago** that concatenates the **Six Months Ago** variable with an "01" (the first day of the month) and the **Year of Six Months Ago** variable. To do this, it must also perform the **FormatNumber** function on the variables because only character data types can be concatenated. This concatenation is nested within a **ToDate** function, which turns the concatenation into a valid, usable date.

   ```
   =ToDate(FormatNumber(<Six Months Ago> ,"00") + "/" + "01" + "/"
   +FormatNumber(<Year of Six Months Ago> ,"0000") ,"mm/dd/yyyy")
   ```

5. Apply a filter on **Invoice Date** with the following definition:

   ```
    =<Invoice Date> Between (<First Day of Six Months Ago> ,
   CurrentDate())
   ```

6. Insert the following formula as a report title:

   ```
   ="Sales Invoices:  " + FormatDate(<First Day of Six Months Ago> ,
   "Mmmm d, yyyy") + " - " + FormatDate(CurrentDate() ,"Mmmm d, yyyy")
   ```

7. Your report should look similar to the one below.

Sales Invoices: August 1, 2002 - February 20, 2003

Sales Person	Invoice Date	Revenue
Fischer	8/22/2002	$23,160.00
Fischer	11/22/2002	$16,380.00
Fischer	2/15/2003	$23,120.00
Galagers	8/17/2002	$24,780.00
Galagers	9/19/2002	$36,654.00
Galagers	11/17/2002	$32,630.00
Galagers	12/19/2002	$38,744.00
Galagers	2/10/2003	$20,495.00
Ishimoto	8/27/2002	$26,040.00
Ishimoto	9/29/2002	$39,442.00
Ishimoto	11/27/2002	$14,380.00
Ishimoto	12/29/2002	$27,136.00
Ishimoto	2/20/2003	$22,960.00

Crosstab Labels

Problem:

I want to label the measure used in the body of a crosstab block, but I do not want the label to appear for each column.

	January	February	March	April
	Sales Revenue	Sales Revenue	Sales Revenue	Sales Revenue
California	$288,260	$161,506	$279,979	$260,420
Colorado	$82,188	$49,653	$72,912	$71,550
DC	$108,675	$73,420	$96,913	$94,272
Florida	$85,677	$42,192	$76,014	$64,080
Illinois	$116,260	$53,410	$85,989	$90,183
Massachusetts	$83,637	$60,773	$75,890	$71,267
New York	$320,715	$163,063	$263,382	$215,974
Texas	$415,955	$259,434	$427,091	$354,584

Solution:

1. Create a crosstab block using **Year**, **Month Name**, and **Sales revenue** from the **eFashion** universe. Apply a condition to retrieve only the current year and a condition that prompts the user for which month(s) to retrieve. Format the crosstab block with **Month Name** at the top and **State** along the left side, and then insert a row below **Month Name**.

2. Create a new variable named **Even Number of Months** to determine if the user has selected an even number of months.

   ```
   =Even(Count(<Month Name>) In Block)
   ```

   Create a new variable named **Odd Number of Months** to determine if the user has selected an odd number of months.

   ```
   =Odd(Count(<Month Name>) In Block)
   ```

3. Create a variable named **Label Placement** that determines the center column of the crosstab block.

   ```
   =((Count(<Month Name>) In Block)+3)/2
   ```

4. Create a variable named **Revenue Label** that displays the text “Sales Revenue” based on the number of months in the crosstab block and the value returned by the **Label Placement** variable.

   ```
   =If (<Even Number of Months> And ColumnNumber()=2) Then "Sales
   Revenue" Else If (<Odd Number of Months> And ColumnNumber()=<Label
   Placement>) Then "Sales Revenue"
   ```

5. Drag **Revenue Label** from the Report Manager and drop it into the row you inserted in step 1 above. You should see a crosstab like the one below if the user selects an odd number of months.

	January	February	March
		Sales Revenue	
California	$288,260	$161,506	$279,979
Colorado	$82,188	$49,653	$72,912
DC	$108,675	$73,420	$96,913
Florida	$85,677	$42,192	$76,014
Illinois	$116,260	$53,410	$85,989
Massachusetts	$83,637	$60,773	$75,890
New York	$320,715	$163,063	$263,382
Texas	$415,955	$259,434	$427,091

You should see a crosstab like the one below if the user selects an even number of months.

	January	February	March	April
	Sales Revenue			
California	$288,260	$161,506	$279,979	$260,420
Colorado	$82,188	$49,653	$72,912	$71,550
DC	$108,675	$73,420	$96,913	$94,272
Florida	$85,677	$42,192	$76,014	$64,080
Illinois	$116,260	$53,410	$85,989	$90,183
Massachusetts	$83,637	$60,773	$75,890	$71,267
New York	$320,715	$163,063	$263,382	$215,974
Texas	$415,955	$259,434	$427,091	$354,584

Graphic Alerters

Problem:

I want graphics to appear as alerters, but I don't want to embed graphics in my document.

Solution:

1. First, open **Character Map** to find appropriate graphics. To do this, click the **Start** button, then select **Programs → Accessories → System Tools → Character Map**.

2. Select **Wingdings** from the drop-down list.

3. In BUSINESSOBJECTS, create a table block using **Year**, **State**, and **Sales revenue** from the **eFashion** universe.

4. Insert a column to the right of **Sales revenue** and type **Rating** as the header for the new column.

5. Click in the body of the new column, and then click the **Alerters** button in the toolbar.

6. Add a new alerter named **Rating** with the following operators and values:



7. Click the drop-down arrow next to the **Result** field for the first condition and select **Text**.

8. Switch to the **Character Map** window and copy the frowning symbol. Then, switch back to BUSINESSOBJECTS and paste the symbol. Click **OK**.

9. Click the drop-down arrow next to the **Result** field for the first condition and select **Format**.

10. On the **Font** tab, select **Wingdings** from the font list, select a large font size, and change the font color.

11. Repeat steps 7-10 for the other two conditions, selecting an appropriate symbol and color for each condition.

12. Click **OK** to close the Edit Alerters dialog box. Apply the **Rating** alerter to the **Rating** column in the table block.

13. Your report should look similar to the one below.

Year	State	Sales revenue	Rating
2002	California	$1,704,211	
2002	Colorado	$448,302	
2002	DC	$693,211	
2002	Florida	$405,985	
2002	Illinois	$737,914	
2002	Massachussetts	$238,819	
2002	New York	$1,667,696	
2002	Texas	$2,199,677	
2003	California	$2,782,680	

Banded Report

Problem:

I want every other row of my report to be shaded.

Solution:

1. Create a new report using **Year**, **Quarter**, **State**, **Sales revenue**, and **Quantity sold** from the **eFashion** universe. Insert a break on **Year**, and then insert a break on **Quarter**. Insert sums for **Sales revenue** and **Quantity sold**.

2. Format the body of the table block to have a **white** fill color.

3. Select the body of the table block and click the **Alerters** button.

4. Create a new alerter named **Shading**.

5. On the **Conditions** tab, click **Add** to create a new variable. Name the new variable **Even Test** and enter the following formula.

   ```
   =Even(LineNumber())
   ```

6. From the **Operator 1** drop-down list, select **= (equal to)**. In the **Value 1** field, type **0**. Format the result to have **black** text and a **gray** fill color.



7. Click **OK** to return to the Alerters dialog box.

8. Be sure the check box for the **Shading** alerter is selected, and then click **OK**.

9. Your report should look similar to the one below.

Sales Report

Year	Quarter	State	Sales revenue	Quantity sold
2002	Q1	California	$519,220	3,509
		Colorado	$131,797	921
		DC	$208,324	1,467
		Florida	$137,530	924
		Illinois	$256,454	1,711
		Massachusetts	$92,596	609
		New York	$555,983	3,717
		Texas	$758,796	5,278
	Q1	**Sum:**	**$2,660,700**	**18,136**
Year	**Quarter**	**State**	**Sales revenue**	**Quantity sold**
	Q2	California	$441,494	2,788
		Colorado	$129,076	828
		DC	$179,863	1,149

Extra: What if I want every third row of my report to be shaded, but I want the shading to look the within each break level?

1. Create a new alerter named **Every 3rd Shading With Reset**.

2. On the **Conditions** tab, click **Add** to create a new variable. Name the new variable **Mod 3 Reset** and enter the following formula.

```
=Mod(RunningCount(LineNumber();<Quarter>),3)
```

3. From the **Operator 1** drop-down list, select **= (equal to)**. In the **Value 1** field, type **0**. Format the result to have **black** text and a **yellow** fill color.

4. From the second **Operator 1** drop-down list, select **= (equal to)**. In the **Value 1** field, type **1**. Format the result to have **black** text and a **blue** fill color.

5. From the third **Operator 1** drop-down list, select **= (equal to)**. In the **Value 1** field, type **2**. Format the result to have **black** text and a **pink** fill color.



6. Click **OK** to return to the Alerters dialog box.

7. Be sure the check box for the **Every Third Shading With Reset** alerter is selected, and then click **OK**.

8. Your report should look similar to the one below.

Year	Quarter	State	Sales revenue	Quantity sold
2002	Q1	California	$519,220	3,509
		Colorado	$131,797	921
		DC	$208,324	1,467
		Florida	$137,530	924
		Illinois	$256,454	1,711
		Massachusetts	$92,596	609
		New York	$555,983	3,717
		Texas	$758,796	5,278
	Q1	**Sum:**	**$2,660,700**	**18,136**
Year	**Quarter**	**State**	**Sales revenue**	**Quantity sold**
	Q2	California	$441,494	2,788
		Colorado	$129,076	828
		DC	$179,863	1,149

Ranking Specific Sections

Problem:

I only want one of my report sections ranked.

Solution:

1. Create a new report using **Resort**, **Year**, and **Number of guests** from the **Island Resorts Marketing** universe.
2. Copy the table that is created and paste next to the original table.
3. Set **Resort** as a master cell.
4. Apply a ranking to the first table to display the **Top 1 Year** based on **Number of guests**.



5. Right-click on the ranked table and select **Format Table**. On the **Appearance** tab, select the **Hide Block** check box and type the following formula:

```
=<Resort> InList ("French Riviera", "Hawaiian Club")
```

6. Click **OK**.

7. Next, right-click on the second table and select **Format Table**. On the **Appearance** tab, select the **Hide Block** check box and type the following formula:

```
=<Resort>="Bahamas Beach"
```

8. Left-align the second table with the first table. Your report should look similar to the one below.

Bahamas Beach

Year	Number of gues
FY02	191.00

French Riviera

Year	Number of gues
FY01	154.00
FY02	147.00
FY03	145.00

Hawaiian Club

Year	Number of gues
FY01	177.00
FY02	187.00
FY03	176.00

Displaying the Last Refreshed Date/Time

Problem:

I want to concatenate the last refreshed date and time with some text, but I get syntax errors when I try to concatenate text with the results of the LastExecutionDate function.

Solution:

1. Open an existing report.

2. Insert a free-standing cell in the report header section.

3. Type the following formula in the free-standing cell, replacing **<Your Object>** with an object from your data provider:

```
="Document Last Refreshed on " + FormatDate(LastExecutionDate
(DataProvider(<Your Object>)) , "Mmm d, yyyy (h:mm AM/PM)")
```

4. You should get a result similar to the one below.